celine_gb.mws

Math 262a, Fall 1999, Glenn Tesler, 12/2/99
Sums via noncommutative Grobner bases

Load Chyzak's software. See his web site for numerous papers with more examples along these lines:

http://pauillac.inria.fr/algo/chyzak/mgfun.html

```
> restart; with(Mgfun): with(Holonomy): with(Groebner):
  with(Ore_algebra):
```

Example 1. Similar to Sister Celine's algorithm

Define a noncommutative algebra with n,k,x, and shifts for n and k

```
> A :=
  skew_algebra(shift=[Sn,n],shift=[Sk,k],comm=[x],polynom=
  [n,k]);
```

$$A := Ore\_algebra$$

Describe the summand binomial(n,k)*x^k by a rectangular system

```
> F := (n,k) -> binomial(n,k)*x^k;
  rect_F := hypergeom_to_dfinite(F(n,k),A);
```

$$F := (n, k) \to \mathrm{binomial}(n, k)\, x^k$$

$$rect\_F := [\, Sn\,(n+1-k) - n - 1,\ Sk\,(k+1) - x\,n + x\,k\,]$$

Term order to eliminate k: to see if k^a1 * Sk^a2 Sn^a3 n^a4 > k^b1 * Sk^b2 Sn^b3 n^b4, first compare k^a1 and k^b1 in grlex order; and if they're equal, break the tie by comparing Sk^a2 Sn^a3 n^a4 and Sk^b2 Sn^b3 n^b4 in grlex (Sk>Sn>n) order. This turns out to be more efficient that using lex order (k>Sk>Sn>n) since we only care to eliminate k, rather than successivly eliminate k; Sk; Sn.

```
> T := termorder(A,lexdeg([k],[Sk,Sn,n]));
```

$$T := term\_order$$

Grobner basis with lower elements free of k (if any elements free of k exist at all)

```
> GB := gbasis(rect_F,T);
```

$$GB := [\, Sn\,Sk + Sk\,Sn\,n - Sk\,n - Sk - x\,n - x,\ -Sn\,n - Sn + Sn\,k + n + 1,\ Sk\,k + Sk - x\,n + x\,k\,]$$

Select the elements free of k

```
> kfree_GB := select(p -> not has(p,k), GB);
```

$$kfree\_GB := [\, Sn\,Sk + Sk\,Sn\,n - Sk\,n - Sk - x\,n - x\,]$$

This says that F(n,k) satisfies the recursion

```
> recop := kfree_GB[1]:
  applyopr(recop,' F'(n,k),A) = 0;
```

$$(-n-1)\,\mathrm{F}(n,k+1)+(n+1)\,\mathrm{F}(n+1,k+1)+(-x\,n-x)\,\mathrm{F}(n,k)=0$$

As with Celine's algorithm, we would now sum this recursion over all k to get f(n)=sum(F(n,k),k=-infinity..infinity). But we are working at the level of recursion operators; substituting Sk -> 1 gives the recursion operator applicable to f(n).

```
> frecop := subs(Sk=1,recop);
```

$$frecop := Sn + Sn\,n - n - 1 - x\,n - x$$

```
> factor(");
```

$$(n+1)\,(Sn-x-1)$$

```
> frecop2:=select(has,",Sn);
```

$$frecop2 := Sn - x - 1$$

Thus, f(n) satisfies

```
> applyopr(frecop2,f(n),A)=0;
```

$$\mathrm{f}(n+1)+(-x-1)\,\mathrm{f}(n)=0$$

which is solved by

```
> rsolve(",f(n));
```

$$\mathrm{f}(0)\,(x+1)^n$$

and then using initial conditions gives the complete answer to the original summation.

```
>
```

Example 2. Double sum

```
> h := (2*k+2*j+n+m)! * (x/4)^(k+j) / ((k+n)! * (j+m)! *
  k! * j!);
```

$$h := \frac{(2\,k+2\,j+n+m)!\,\left(\frac{1}{4}\,x\right)^{(k+j)}}{(k+n)!\,(j+m)!\,k!\,j!}$$

We want to compute

```
> H(n,m,x) = Sum(Sum(h,j=0..infinity),k=0..infinity);
```

$$\mathrm{H}(n,m,x)=\sum_{k=0}^{\infty}\left(\sum_{j=0}^{\infty}\frac{(2\,k+2\,j+n+m)!\,\left(\frac{1}{4}\,x\right)^{(k+j)}}{(k+n)!\,(j+m)!\,k!\,j!}\right)$$

and as with Celine's algorithm, we will do so by finding a recurrence/diffeq satisfied by the summand, with the coefficients of the recurrence/diffeq free of j and k.

```
> A :=
  skew_algebra(diff=[Dx,x],shift=[Sk,k],shift=[Sj,j],comm=
  [n,m],polynom={k,j});
```

$$A := Ore\_algebra$$

Term order to eliminate j and k:

```
> T := termorder(A,lexdeg([k,j],[Sj,Sk,Dx]));
```

$$T := term\_order$$

```
> rect_h := hypergeom_to_dfinite(h,A);
```

$$rect\_h := [(4k^2+4kn+4n+8k+4)\,Sk - 8xkj - 4xkm - 4xj^2 - 4xjn - 4xjm - 2x - 3nx - 6kx - 6xj - 4xkn - xn^2 - 3xm - 4xk^2 - 2xnm - xm^2,\ Dx\,x - k - j,\ (4j^2+8j+4+4jm+4m)\,Sj - 8xkj - 4xkm - 4xj^2 - 4xjn - 4xjm - 2x - 3nx - 6kx - 6xj - 4xkn - xn^2 - 3xm - 4xk^2 - 2xnm - xm^2]$$

```
> GB := gbasis(rect_h,T):
```

I suppressed printing the output because it is so large! There are

```
> nops(GB);
```

$$10$$

different polynomials in it. The number of terms, and the particular variables, occurring in each are

```
> map(p->[nops(p),indets(p)],GB);
```

$$[[308, \{x, n, Sk, Dx, m, Sj\}], [512, \{x, n, Sk, Dx, m, Sj\}], [3, \{x, k, j, Dx\}], [73, \%1], [28, \%1], [193, \%1], [104, \%1], [468, \%1], [15, \{x, n, k, Sk, Dx, m\}], [16, \{x, n, k, Dx, m, Sj\}]]$$

$$\%1 := \{x, n, k, Sk, Dx, m, Sj\}$$

We see the first two expressions are free of k and j. The other expressions are irrelevant. But to give an idea what a Grobner basis is like (output is MUCH larger than the input), we show its small elements:

```
> prettypol := p -> collect(p,[Dx,Sk,Sj],factor);
```

$$prettypol := p \to \mathrm{collect}(p, [Dx, Sk, Sj], factor)$$

```
> for r in [3,4,5,9,10] do print(`
  GB`[r]=prettypol(GB[r])) od;
```

$$GB_3 = -Dx\,x + k + j$$

$$GB_4 = ((-8x^4 + 8\,Sj\,x^3)\,Sk - 24x^4\,Sj)\,Dx^3 + ((12x^2(n+m+2)\,Sj - 4x^3(5n+3m+15+4k))\,Sk - 4x^3(29-4k+7n+9m)\,Sj)\,Dx^2 + ((4x(m+1)(m+2+3n)\,Sj - 2x^2(10nm+20k+24m+3m^2+7n^2+34n+8km+42+8kn))\,Sk - 2x^2(30n+40m-20k+5n^2-8kn+9m^2-8km+14nm+46)\,Sj)\,Dx +$$

$$(4\,(n-m)\,(n+m)\,(k+1)\,Sj - x\,(n+m+2)\,(n+m+1)\,(3\,n+m+4\,k+6))\,Sk$$
$$-\,x\,(n+m+2)\,(n+m+1)\,(n+3\,m-4\,k+2)\,Sj$$

$$GB_5 = ((-4\,Sj\,x^2+4\,x^3)\,Sk - 4\,Sj\,x^3)\,Dx^2$$
$$+\,((4\,x\,(1-m+2\,k)\,Sj + 2\,x^2\,(5+2\,n+2\,m))\,Sk - 2\,x^2\,(5+2\,n+2\,m)\,Sj)\,Dx$$
$$+\,(4\,(k+1)\,(n+m)\,Sj + x\,(n+m+2)\,(n+m+1))\,Sk$$
$$-\,x\,(n+m+2)\,(n+m+1)\,Sj$$

$$GB_9 = -4\,Dx^2\,x^3 - 2\,x^2\,(5+2\,n+2\,m)\,Dx + 4\,(k+1)\,(k+n+1)\,Sk$$
$$-\,x\,(n+m+2)\,(n+m+1)$$

$$GB_{10} = (-4\,x^3+4\,Sj\,x^2)\,Dx^2 + (-4\,x\,(2\,k-1-m)\,Sj - 2\,x^2\,(5+2\,n+2\,m))\,Dx$$
$$+\,4\,k\,(k-m)\,Sj - x\,(n+m+2)\,(n+m+1)$$

Select the operators that are free of k and j (i.e., GB[1] and GB[2], not printed above):

```
> GB_no_kj := select(p -> not has(p,{k,j}), GB):
> nops(GB_no_kj);
```

$$2$$

```
> indets(GB_no_kj[1]), indets(GB_no_kj[2]);
```

$$\{x, n, Sk, Dx, m, Sj\}, \{x, n, Sk, Dx, m, Sj\}$$

Let H = sum_{k,j} h. If h satisfies

C(x,n,m,Dx,Sk,Sj) * h(x,n,m,k,j) = 0

then summing this over all k,j gives

C(x,n,m,Dx,1,1) * H(x,n,m) = 0

i.e., a differential equation w.r.t. x that H satisfies. We have two such operators C.

```
> CT := subs(Sk=1,Sj=1,GB_no_kj): map(prettypol,CT);
```

$$[-32\,x^4\,(-1+4\,x)\,Dx^5 - 16\,x^3\,(-5\,m-5\,n+20\,n\,x-18+104\,x+20\,x\,m)\,Dx^4 -$$
$$32\,x^2\,(-2\,n^2+10\,x\,n^2+20\,x\,n\,m+185\,x-19-5\,n\,m-13\,n+10\,x\,m^2+84\,x\,m$$
$$-\,13\,m+84\,n\,x-2\,m^2)\,Dx^3 - 16\,x\,(323\,x\,m+10\,x\,n^3-n^3+96\,x\,n^2-16-23\,m$$
$$-\,23\,n+96\,x\,m^2-10\,n^2+378\,x+10\,x\,m^3-m^3-24\,n\,m+30\,x\,n\,m^2+192\,x\,n\,m$$
$$-\,6\,n\,m^2-6\,n^2\,m+30\,m\,x\,n^2-10\,m^2+323\,n\,x)\,Dx^2 + (32\,m^2\,n^2+32\,m^2-1152\,x$$
$$+\,16\,n+64\,n\,m-1904\,n\,x+32\,n^2-1056\,x\,n\,m^2-1056\,m\,x\,n^2-352\,x\,m^3$$
$$-\,352\,x\,n^3-160\,n\,x\,m^3-40\,n^4\,x-240\,x\,m^2\,n^2-160\,x\,m\,n^3-40\,x\,m^4-1208\,x\,n^2$$
$$+\,16\,m+16\,n^3+16\,m^3+64\,n\,m^2+64\,n^2\,m-1904\,x\,m-2416\,x\,n\,m-1208\,x\,m^2$$
$$+\,16\,m^3\,n+16\,n^3\,m)\,Dx - 4\,(n+m)\,(n+m+2)^2\,(n+m+1)^2,$$
$$-16\,x^3\,(n-m)\,(n+m)\,(-1+4\,x)\,Dx^4$$
$$-\,32\,x^2\,(n-m)\,(n+m)\,(-n+4\,n\,x+18\,x-3-m+4\,x\,m)\,Dx^3 - 16\,x\,(n-m)$$

$(n+m)\,($
$6\,x\,n^2 - n^2 - 6\,n + 12\,x\,n\,m - 3\,n\,m + 42\,n\,x + 77\,x - 7 - 6\,m + 42\,x\,m + 6\,x\,m^2 - m^2$
$)\,Dx^2 - 16\,(n-m)\,(n+m)\,(2\,x\,n^3 + 15\,x\,n^2 + 6\,m\,x\,n^2 - n^2\,m - n^2 - 3\,n\,m - 2\,n$
$+ 6\,x\,n\,m^2 + 30\,x\,n\,m + 39\,n\,x - n\,m^2 - 1 + 39\,x\,m + 35\,x + 2\,x\,m^3 - 2\,m + 15\,x\,m^2$
$- m^2)\,Dx - 4\,(n-m)\,(n+m)\,(n+m+2)^2\,(n+m+1)^2\,]$

Since we have two differential equations w.r.t. x that H satisfies, we take their GCD:

```
> GCD := skew_gcdex(op(CT),Dx,A):  GCD := GCD[1]:
  prettypol(GCD);
```

$-4\,x^3\,(n-m)\,(n+m)\,(-1+4\,x)\,Dx^4$
$- 8\,x^2\,(n-m)\,(n+m)\,(-n + 4\,n\,x + 18\,x - 3 - m + 4\,x\,m)\,Dx^3 - 4\,x\,(n-m)$
$(n+m)\,($
$6\,x\,n^2 - n^2 - 6\,n + 12\,x\,n\,m - 3\,n\,m + 42\,n\,x + 77\,x - 7 - 6\,m + 42\,x\,m + 6\,x\,m^2 - m^2$
$)\,Dx^2 - 4\,(n-m)\,(n+m)\,(2\,x\,n^3 + 15\,x\,n^2 + 6\,m\,x\,n^2 - n^2\,m - n^2 - 3\,n\,m - 2\,n$
$+ 6\,x\,n\,m^2 + 30\,x\,n\,m + 39\,n\,x - n\,m^2 - 1 + 39\,x\,m + 35\,x + 2\,x\,m^3 - 2\,m + 15\,x\,m^2$
$- m^2)\,Dx - (n-m)\,(n+m)\,(n+m+2)^2\,(n+m+1)^2$

For n <> +/- m, we can factor out (n-m)(n+m):

```
> GCD2 := simplify(GCD / ((n-m)*(n+m))): prettypol(GCD2);
```

$-4\,x^3\,(-1+4\,x)\,Dx^4 - 8\,x^2\,(-n + 4\,n\,x + 18\,x - 3 - m + 4\,x\,m)\,Dx^3 - 4\,x\,($
$6\,x\,n^2 - n^2 - 6\,n + 12\,x\,n\,m - 3\,n\,m + 42\,n\,x + 77\,x - 7 - 6\,m + 42\,x\,m + 6\,x\,m^2 - m^2$
$)\,Dx^2 + (-156\,n\,x + 8\,n - 140\,x + 4\,n\,m^2 - 60\,x\,m^2 + 4\,n^2\,m - 8\,x\,m^3 + 4\,n^2 + 12\,n\,m$
$- 156\,x\,m + 8\,m - 24\,m\,x\,n^2 - 24\,x\,n\,m^2 - 120\,x\,n\,m + 4 - 60\,x\,n^2 - 8\,x\,n^3 + 4\,m^2)$
$Dx - (n+m+2)^2\,(n+m+1)^2$

In normal notation, here is the differential equation to solve:

```
> applyopr(GCD2,H(x),A) = 0;
```

$(-6\,m^2\,n^2 - 26\,n\,m - 4 - 13\,m^2 - 12\,n - 12\,m - n^4 - 6\,m^3 - 13\,n^2 - 18\,n^2\,m - 6\,n^3$
$- m^4 - 4\,n^3\,m - 18\,n\,m^2 - 4\,m^3\,n)\,\mathrm{H}(x) + (-156\,n\,x + 8\,n - 140\,x + 4\,n\,m^2$
$- 60\,x\,m^2 + 4\,n^2\,m - 8\,x\,m^3 + 4\,n^2 + 12\,n\,m - 156\,x\,m + 8\,m - 24\,m\,x\,n^2 - 24\,x\,n\,m^2$
$- 120\,x\,n\,m + 4 - 60\,x\,n^2 - 8\,x\,n^3 + 4\,m^2)\left(\frac{\partial}{\partial x}\mathrm{H}(x)\right) + (28\,x - 24\,x^2\,m^2 + 24\,x\,m$
$+ 24\,n\,x - 308\,x^2 - 48\,x^2\,n\,m - 168\,n\,x^2 - 168\,x^2\,m + 4\,x\,m^2 + 4\,x\,n^2 - 24\,x^2\,n^2$
$+ 12\,x\,n\,m)\left(\frac{\partial^2}{\partial x^2}\mathrm{H}(x)\right)$

$$+ (8 x^2 m - 144 x^3 - 32 x^3 m - 32 x^3 n + 8 n x^2 + 24 x^2) \left(\frac{\partial^3}{\partial x^3} \mathrm{H}(x) \right)$$

$$+ (4 x^3 - 16 x^4) \left(\frac{\partial^4}{\partial x^4} \mathrm{H}(x) \right) = 0$$

```
> Heq := ":
```

It’s fourth order, so there are 4 linearly independent solutions.. We can solve it by Laurent series, and then it is expressed as a single summation instead of a double summation; our solution will be a linear combination of 4 Laurent series.

```
> Hlaurent := sum(b[k]*x^(k+alpha),k=0..infinity);
```

$$Hlaurent := \sum_{k=0}^{\infty} b_k \, x^{(k+\alpha)}$$

Applying the operator termwise gives

```
> termwise :=
  collect(expand(applyopr(GCD2,b[k]*x^(k+alpha),A)),x,fact
  or);
```

$$termwise := -b_k \, x^k \, x^\alpha \, (n + 2 + 2k + 2\alpha + m)^2 \, (n + 1 + 2k + 2\alpha + m)^2$$

$$+ 4 \frac{b_k \, x^k \, x^\alpha \, (k + \alpha)(k + \alpha + m)(k + \alpha + n)(n + k + \alpha + m)}{x}$$

Shift the index in the second term to collect the coeff. of x^(k+alpha) in the sum:

```
> xkcoeff :=
  factor(simplify(op(1,termwise)/x^(k+alpha)))+factor(simp
  lify(subs(k=k+1,op(2,termwise))/x^(k+alpha)));
```

$$xkcoeff := -b_k \, (n + 2 + 2k + 2\alpha + m)^2 \, (n + 1 + 2k + 2\alpha + m)^2$$

$$+ 4 \, b_{k+1} \, (k + 1 + \alpha)(k + 1 + \alpha + m)(k + 1 + \alpha + n)(n + k + 1 + \alpha + m)$$

So b[k+1]/b[k] =

```
> bratio := factor(solve(xkcoeff,b[k+1])/b[k]);
```

$$bratio := \frac{1}{4} \frac{(n + 2 + 2k + 2\alpha + m)^2 \, (n + 1 + 2k + 2\alpha + m)^2}{(k + 1 + \alpha)(k + 1 + \alpha + m)(k + 1 + \alpha + n)(n + k + 1 + \alpha + m)}$$

We find alpha using b[k]=0 for k<0 but b[0]<>0. This gives the indicial equation

```
> subs(k=-1,xkcoeff)=0;
```

$$-b_{-1} \, (n + 2\alpha + m)^2 \, (n - 1 + 2\alpha + m)^2 + 4 \, b_0 \, \alpha \, (\alpha + m)(\alpha + n)(n + \alpha + m) = 0$$

```
> subs(b[-1]=0,");
```

$$4 \, b_0 \, \alpha \, (\alpha + m)(\alpha + n)(n + \alpha + m) = 0$$

```
> solve(",alpha);
```

$$-m, 0, -n - m, -n$$

Thus there are 4 different Laurent series solutions to the equation, beginning with terms x^(-m), x^(-n), x^(-n-m), x^0. It’s a 4th order equation, so these span all solutions. The original double sum given clearly begins with x^0, so we may dispose of the other solutions (provided n,m<>0 and n<>-m). Then the term ratio for the Taylor series we want is

```
> bxratio := subs(alpha=0,bratio)*x;
```

$$bxratio := \frac{1}{4}\frac{(n+2+2k+m)^2(n+1+2k+m)^2 x}{(k+1)(k+1+m)(k+1+n)(n+k+1+m)}$$

Note that if n or m = 0, or n=-m, the usual method of solving this differential equation would give additional solutions with logarithms, which would still be linearly independent of this solution we are developing, so this solution is actually valid even in these other cases.

Also in the original double sum, the coefficient of x^0 is

```
> x0coeff := subs(k=0,j=0,h);
```

$$x0coeff := \frac{(n+m)!}{n!\,m!}$$

So we have found the Taylor series using hypergeometric functions.

```
> Hhyper := x0coeff * hypergeom(['(n+m+2)/2'$2,
  '(n+m+1)/2'$2], [m+1,n+1,n+m+1], 2^2 * 2^2 * x/4);
```

$$Hhyper := (n+m)!\,\mathrm{hypergeom}\left(\left[\frac{1}{2}n+\frac{1}{2}m+1, \frac{1}{2}n+\frac{1}{2}m+1, \frac{1}{2}n+\frac{1}{2}m+\frac{1}{2}, \frac{1}{2}n+\frac{1}{2}m+\frac{1}{2}\right], [n+1, n+m+1, m+1], 4x\right)/(n!\,m!)$$

Normal notation:

```
> read gmisc: # routine to put it in usual notation
> ``(x0coeff) *
  fmat_pFq(['(n+m)/`2`+1'$2,'(n+m+1)/`2`'$2],[m+1,n+1,n+m+
  1], 4*x);
```

$$\left(\frac{(n+m)!}{n!\,m!}\right)\, {}_4F_3\left[\begin{matrix} \frac{n+m}{2}+1, & \frac{n+m}{2}+1, & \frac{n+m+1}{2}, & \frac{n+m+1}{2} \\ m+1, & n+1, & n+m+1 & \end{matrix} ;\ 4x\right]$$

Example 3: q-sums

```
> restart; read(gmisc);
```

Chyzak's software

```
> with(Mgfun): with(Holonomy): with(Groebner):
  with(Ore_algebra):
```

Chyzak's q-functions are different than Koepf's; Koepf's are more powerful, let's use them instead.

```
> with(qsum);
```

Copyright 1998, Harald Boeing & Wolfram Koepf

Konrad-Zuse-Zentrum Berlin

[*qsum/init*]

This is the summand of the q-analogue of the Chu-Vandermonde identity (Koepf p. 60 # 4.18; p. 158)

```
> h := qphihyperterm([q^(-n),b],[c],q,c*q^n/b,k);
```

$$h := \frac{\text{qpochhammer}(q^{(-n)}, q, k)\ \text{qpochhammer}(b, q, k) \left(\frac{c\, q^n}{b}\right)^k}{\text{qpochhammer}(c, q, k)\ \text{qpochhammer}(q, q, k)}$$

The annihilators are generated by these ones. Let qn denote q^n and qk denote q^k, and clear denominators.

```
> rect_h := [Sn - qsimplify(subs(n=n+1,h)/h), Sk -
  qsimplify(subs(k=k+1,h)/h) ];
  rect_h := subs(q^k=qk,q^n=qn,rect_h);
  rect_h := map(numer,rect_h);
```

$$rect\_h := \left[Sn - \frac{(q^n\, q - 1)\, q^k}{q^n\, q - q^k},\ Sk - \frac{(-q^n + q^k)\,(-1 + b\, q^k)\, c}{b\,(-1 + c\, q^k)\,(-1 + q\, q^k)} \right]$$

$$rect\_h := \left[Sn - \frac{(qn\, q - 1)\, qk}{qn\, q - qk},\ Sk - \frac{(-qn + qk)\,(-1 + b\, qk)\, c}{b\,(-1 + c\, qk)\,(-1 + q\, qk)} \right]$$

$$rect\_h := [Sn\, qn\, q - Sn\, qk - qk\, qn\, q + qk,$$
$$Sk\, b - Sk\, b\, q\, qk - Sk\, b\, c\, qk + Sk\, b\, c\, qk^2\, q - c\, qn + c\, qn\, b\, qk + c\, qk - c\, b\, qk^2]$$

Set up an Ore algebra for these operators: Sn H(q^n, q^k) = H(q*q^n, q^k), etc.

```
> A :=
  skew_algebra(qdilat=[Sn,qn=q^n],qdilat=[Sk,qk=q^k],comm=
  [q,b,c],polynom=[qn,qk]);
```

$$A := Ore\_algebra$$

Term order to eliminate k (in the form qk=q^k):

```
> T := termorder(A,lexdeg([qk],[Sk,Sn,qn]));
```

$$T := term\_order$$

```
> GB := gbasis(rect_h,T);
```

$$GB := [-Sn\,Sk\,b - q^3\,qn\,Sk\,b + q^4\,qn^2\,Sk\,b + q^3\,qn\,Sn\,Sk\,b - q^4\,b\,Sk\,Sn\,qn^2 - q^3\,c\,b\,Sk\,Sn\,qn^2 + Sn\,c\,qn\,q - q\,Sn\,Sk\,b - q^2\,qn\,Sk\,b + c\,b\,Sk\,qn^2\,q^3\,Sn^2 - c\,b\,Sk\,Sn^2\,qn\,q - b\,Sk\,Sn^2\,qn\,q^2 - Sn\,q^3\,c\,qn^2 + q\,c\,b\,Sk\,Sn\,qn + Sn^2\,Sk\,b + q^4\,c\,b\,qn^3\,Sn - b\,c\,q^2\,Sn\,qn^2 + 2\,b\,Sk\,Sn\,qn\,q^2 - q^4\,c\,qn^3 + q^3\,c\,qn^2 - q\,c\,qn + q\,Sk\,b + q^2\,c\,qn^2, -Sn\,qn\,q + Sn\,qk + qk\,qn\,q - qk, c\,b\,Sk\,qn^2\,q^2\,qk - q\,qn\,Sk\,b\,c\,qk - q\,c\,qn^2\,b\,qk + c\,qn\,b\,qk - c\,b\,Sk\,qn^2\,Sn\,q + c\,b\,Sk\,Sn\,qn + b\,Sk\,Sn\,qn\,q - Sn\,Sk\,b - q\,qn\,Sk\,b + q\,c\,qn^2 + Sk\,b - c\,qn, Sk\,b - Sk\,b\,q\,qk - Sk\,b\,c\,qk + Sk\,b\,c\,qk^2\,q - c\,qn + c\,qn\,b\,qk + c\,qk - c\,b\,qk^2]$$

```
> prettypol := p ->
  sort(collect(p,{Sn,Sk},factor),[n,k,b,c,Sn,Sk],tdeg);
```

$$prettypol := p \rightarrow \mathrm{sort}(\mathrm{collect}(p, \{Sn, Sk\}, factor), [n, k, b, c, Sn, Sk], tdeg)$$

```
> for r from 1 to nops(GB) do print(‘ GB‘[r] =
  prettypol(GB[r])) od;
```

$$GB_1 = (-1 + q^2\,qn)\,(q\,qn\,c - 1)\,b\,Sn^2\,Sk + q\,(-1 + q^2\,qn)\,(qn\,q - 1)\,b\,Sk - q\,qn\,(-1 + q^2\,qn)\,(qn\,q - 1)\,c + (-(-1 + q^2\,qn)\,(q\,qn\,c + q^2\,qn - q - 1)\,b\,Sk + q\,qn\,(-1 + q^2\,qn)\,(q\,qn\,b - 1)\,c)\,Sn$$

$$GB_2 = (-qn\,q + qk)\,Sn + qk\,(qn\,q - 1)$$

$$GB_3 = -(qn\,c - 1)\,(qn\,q - 1)\,b\,Sn\,Sk + (qn\,q - 1)\,(q\,qn\,qk\,c - 1)\,b\,Sk - qn\,(qk\,b - 1)\,(qn\,q - 1)\,c$$

$$GB_4 = (qk\,c - 1)\,(q\,qk - 1)\,b\,Sk - (-qn + qk)\,(qk\,b - 1)\,c$$

Only the first one is free of qk (=q^k, the only way k enters into this):

```
> GB_no_qk := select(p -> not has(p,qk),GB);
```

$$GB\_no\_qk := [-Sn\,Sk\,b - q^3\,qn\,Sk\,b + q^4\,qn^2\,Sk\,b + q^3\,qn\,Sn\,Sk\,b - q^4\,b\,Sk\,Sn\,qn^2 - q^3\,c\,b\,Sk\,Sn\,qn^2 + Sn\,c\,qn\,q - q\,Sn\,Sk\,b - q^2\,qn\,Sk\,b + c\,b\,Sk\,qn^2\,q^3\,Sn^2 - c\,b\,Sk\,Sn^2\,qn\,q - b\,Sk\,Sn^2\,qn\,q^2 - Sn\,q^3\,c\,qn^2 + q\,c\,b\,Sk\,Sn\,qn + Sn^2\,Sk\,b + q^4\,c\,b\,qn^3\,Sn - b\,c\,q^2\,Sn\,qn^2 + 2\,b\,Sk\,Sn\,qn\,q^2 - q^4\,c\,qn^3 + q^3\,c\,qn^2 - q\,qn\,c + q\,Sk\,b + q^2\,c\,qn^2]$$

```
> nops(GB_no_qk);
```

$$1$$

So we have a k-free (rather q^k - free) recursion for the summand F(n,k).

Let f(n) = sum_k F(n,k). Clearly sum_k Sk^r F(n,k) = sum_k F(n,k) = f(n) too, so summing this recursion over k gives a recursion for f(n).
In operator form, we just set Sk to 1:

```
> f_recops := subs(Sk=1,GB_no_qk);
```

$$f\_recops := [-Sn\, b - q^3\, qn\, b + q^4\, qn^2\, b + q^3\, qn\, Sn\, b - q^4\, b\, Sn\, qn^2 - c\, q^3\, b\, Sn\, qn^2 + Sn\, c\, qn\, q - q\, Sn\, b - q^2\, qn\, b + c\, b\, qn^2\, q^3\, Sn^2 - c\, b\, Sn^2\, qn\, q - b\, Sn^2\, qn\, q^2 - Sn\, q^3\, c\, qn^2 + q\, c\, b\, Sn\, qn + Sn^2\, b + q^4\, c\, b\, qn^3\, Sn - b\, c\, q^2\, Sn\, qn^2 + 2\, b\, Sn\, qn\, q^2 - q^4\, c\, qn^3 + q^3\, c\, qn^2 - q\, qn\, c + b\, q + q^2\, c\, qn^2]$$

```
> factor(f_recops);
```

$$[(-1 + q^2\, qn)\,(b\, c\, q^2\, Sn\, qn^2 - q^2\, c\, qn^2 + q^2\, qn\, b - b\, Sn\, qn\, q^2 + q\, qn\, c - Sn\, c\, qn\, q - q\, c\, b\, Sn\, qn + c\, b\, Sn^2\, qn\, q - b\, q + q\, Sn\, b - Sn^2\, b + Sn\, b)]$$

```
> f_recops := simplify(f_recops[1] / (-1 + qn*q^2));
```

$$f\_recops := b\, c\, q^2\, Sn\, qn^2 - q^2\, c\, qn^2 + q^2\, qn\, b - b\, Sn\, qn\, q^2 + q\, qn\, c - Sn\, c\, qn\, q - q\, c\, b\, Sn\, qn + c\, b\, Sn^2\, qn\, q - b\, q + q\, Sn\, b - Sn^2\, b + Sn\, b$$

```
> prettypol(");
```

$$(q\, qn\, c - 1)\, b\, Sn^2 + (q^2\, qn^2\, b\, c - q\, qn\, b\, c + q\, b + b - q^2\, qn\, b - q\, qn\, c)\, Sn + q\,(b - qn\, c)\,(qn\, q - 1)$$

Thus, f(n) = f(n,q) satisfies the recursion:

```
> qrec := applyopr(f_recops,f(n),A):  qrec = 0;
```

$$(-b + q\, q^n\, b\, c)\, \mathrm{f}(n+2) + (q^2\,(q^n)^2\, b\, c - q\, q^n\, b\, c + q\, b + b - q^2\, q^n\, b - q\, q^n\, c)\, \mathrm{f}(n+1) + (q\, q^n\, c - q^2\, c\,(q^n)^2 + q^2\, q^n\, b - q\, b)\, \mathrm{f}(n) = 0$$

The hypergeometric solutions of this are found by the q-analogue of Petkovsek's Hyper; this is in Koepf's package:

```
> qsol := qrecsolve(qrec,q,f(n),return=qhypergeometric);
  qsol := qsol[1][1]:
```

$$qsol := \left[\left[\frac{\mathrm{qpochhammer}\left(\frac{c}{b}, q, n\right)}{\mathrm{qpochhammer}(c, q, n)}, 0 \le n\right]\right]$$

It only found one, and the recursion was 2nd order. We will find that a multiple of this satisfies two initial conditions, but were that not so, we could use reduction of order to find a second solution..
The summand h has finite support (k=0,1,...,n), so it's easy to compute some values:

```
> sumh := N ->
  factor(qsimplify(sum(subs(n=N,k=K,h),K=0..N)));
```

```
qsol_n := N -> factor(qsimplify(subs(n=N,qsol)));
```

$$sumh := N \to \mathrm{factor}\left(\mathrm{qsimplify}\left(\sum_{K=0}^{N} \mathrm{subs}(n=N, k=K, h)\right)\right)$$

$$qsol\_n := N \to \mathrm{factor}(\mathrm{qsimplify}(\mathrm{subs}(n=N, qsol)))$$

```
> '[sumh(N),qsol_n(N)]'$N=0..3;
```

$$[1, 1], \left[-\frac{b-c}{(-1+c)\,b}, -\frac{b-c}{(-1+c)\,b}\right],$$

$$\left[-\frac{(b-c)\,(-b+q\,c)}{(-1+q\,c)\,(-1+c)\,b^2}, -\frac{(b-c)\,(-b+q\,c)}{(-1+q\,c)\,(-1+c)\,b^2}\right],$$

$$\left[-\frac{(b-c)\,(-b+q\,c)\,(-b+c\,q^2)}{(-1+c\,q^2)\,(-1+q\,c)\,(-1+c)\,b^3}, -\frac{(b-c)\,(-b+q\,c)\,(-b+c\,q^2)}{(-1+c\,q^2)\,(-1+q\,c)\,(-1+c)\,b^3}\right]$$

We don't even have to rescale this solution: it satisfies the first two initial conditions (and we have redundantly checked more) already! So we have proved the q-analogue of the Chu-Vandermonde identity

```
> fmat_pphiq([q^(-n),b],[c],q,c*q^n/b) = '';
  Sum(h,k=0..infinity) = qsol;
```

$$ {}_2\phi_1\left[\begin{matrix} q^{(-n)} \;,\; b \\ \;;\; q \;,\; \dfrac{c\,q^n}{b} \\ c \end{matrix}\right] =$$

$$\sum_{k=0}^{\infty} \frac{\mathrm{qpochhammer}(q^{(-n)}, q, k)\,\mathrm{qpochhammer}(b, q, k)\left(\dfrac{c\,q^n}{b}\right)^k}{\mathrm{qpochhammer}(c, q, k)\,\mathrm{qpochhammer}(q, q, k)} =$$

$$\frac{\mathrm{qpochhammer}\left(\dfrac{c}{b}, q, n\right)}{\mathrm{qpochhammer}(c, q, n)}$$

Example 4. Legendre polynomials (from Chyzak's papers)

Here are three operators that annihilate the Legendre Polynomials
(see Chyzak's paper Holonomic systems and automatic proofs of identities, pp. 24ff & 48ff; and the Introduction in Koepf's book).

```
> DE := (1-x^2)*Dx^2 - 2*x*Dx+n*(n+1);
  RE := (n+2)*Sn^2 - (2*n+3) * x * Sn + (n+1);
```

```
RDE := (1-x^2)*Dx*Sn + (n+1)*x*Sn - (n+1);
```

$$DE := (1 - x^2)\, Dx^2 - 2\, x\, Dx + n\, (n + 1)$$

$$RE := (n + 2)\, Sn^2 - (2\, n + 3)\, x\, Sn + n + 1$$

$$RDE := (1 - x^2)\, Dx\, Sn + (n + 1)\, x\, Sn - n - 1$$

We use Grobner bases **(a)** to derive RE from the other two, and then **(b)** to derive DE from the other two.

(a) create elimination term order to remove Dx from two of these

```
> U := termorder(A,lexdeg([Dx],[n,x,Sn]));
Error, (in termorder) rational indeterminates not allowed to build a term order
```

Redefine the algebra so that only polynomials in x,n are allowed, rather than rational functions

```
> A_poly:=skew_algebra(diff=[Dx,x],shift=[Sn,n],polynom=[x
  ,n]);
```

$$A\_poly := Ore\_algebra$$

(Page 25) Get recurrence equation by taking the diff eq and the mixed diff eq/recurrence eq, and eliminating Dx from it.

```
> U := termorder(A_poly,lexdeg([Dx],[n,x,Sn]));
```

$$U := term\_order$$

```
> bas := map(expand,[DE,RDE]);
```

$$bas := [Dx^2 - Dx^2\, x^2 - 2\, x\, Dx + n^2 + n,\ Dx\, Sn - Dx\, Sn\, x^2 + x\, Sn\, n + x\, Sn - n - 1]$$

```
> GBR := gbasis(bas,U);
```

$$GBR := [-Sn^2\, n^2 - 4\, Sn^2\, n - 4\, Sn^2 + 7\, x\, Sn\, n + 6\, x\, Sn + 2\, x\, Sn\, n^2 - 3\, n - 2 - n^2,$$
$$-Dx\, Sn\, n + n^2 + 2\, n + x\, Dx\, n + x\, Dx - Dx\, Sn + 1,$$
$$-Dx\, Sn\, x - Dx\, n - Dx - 6\, Sn\, n - 5\, Sn - 2\, Sn\, n^2 + Dx\, Sn^2\, n + 2\, Dx\, Sn^2,$$
$$-Dx\, Sn + Dx\, Sn\, x^2 - x\, Sn\, n - x\, Sn + n + 1,\ -Dx^2 + Dx^2\, x^2 + 2\, x\, Dx - n^2 - n]$$

```
> noDx := select(f->not has(f,Dx), GBR);
```

$$noDx := [-Sn^2\, n^2 - 4\, Sn^2\, n - 4\, Sn^2 + 7\, x\, Sn\, n + 6\, x\, Sn + 2\, x\, Sn\, n^2 - 3\, n - 2 - n^2]$$

```
> nops(noDx);
```

$$1$$

```
> factor(noDx[1]);
```

$$(n + 2)\, (-Sn^2\, n + 2\, x\, Sn\, n - n - 2\, Sn^2 - 1 + 3\, x\, Sn)$$

Notes:

1. factor is Maple's built-in command for COMMUTATIVE factorization. In general it is not meaningful for a noncommutative polynomial because the multiplication rules are different. HOWEVER, if we obtain a result a(n) b(n,Sn) c(Sn) then the commutative and noncommutative multiplication rules coincide, so it's valid (where

b(n,Sn) is interpreted as usual with the n's on the left and Sn's on the right, no matter how Maple chooses to display it).
2. We can force the n's on the left, Sn's on the right, as follows:

```
> sort(",[n,x,Sn,Dx],plex);
```

$$(n+2)\,(2\,n\,x\,Sn - n\,Sn^2 - n + 3\,x\,Sn - 2\,Sn^2 - 1)$$

```
> op(2,");
```

$$2\,n\,x\,Sn - n\,Sn^2 - n + 3\,x\,Sn - 2\,Sn^2 - 1$$

```
> collect(",Sn);
```

$$(-2-n)\,Sn^2 + (2\,x\,n + 3\,x)\,Sn - n - 1$$

This is the pure recurrence equation RE above! (Rather, its negative.)

```
>
>
```

(b) Likewise, create an elimination order to remove Sn from 2 equations

```
> V := termorder(A_poly,lexdeg([Sn],[x,n,Dx]));
```

$$V := term\_order$$

```
> bas2 := map(expand,[RE,RDE]);
```

$bas2 :=$

$$[n\,Sn^2 + 2\,Sn^2 - 2\,n\,x\,Sn - 3\,x\,Sn + n + 1,\ Dx\,Sn - Dx\,Sn\,x^2 + n\,x\,Sn + x\,Sn - n - 1]$$

```
> GBD := gbasis(bas2,V);
```

$$GBD := [-2\,x\,Dx\,n - x^2\,n + n + Dx^2\,n + 2\,n^2 - 2\,x^2\,Dx^2\,n - 2\,x\,Dx + Dx^2 + n^3$$
$$- 2\,Dx^2\,x^2 + 2\,Dx\,x^3 - 2\,x^2\,n^2 + 2\,Dx\,x^3\,n + Dx^2\,x^4\,n + Dx^2\,x^4 - x^2\,n^3,$$
$$Sn\,n^2 + 2\,Sn\,n - Dx\,n\,x^2 - Dx\,x^2 - x\,n^2 - 2\,x\,n + Sn - x + Dx\,n + Dx,$$
$$-Dx\,Sn + Dx\,Sn\,x^2 - n\,x\,Sn - x\,Sn + n + 1,\ n\,Sn^2 + 2\,Sn^2 - 2\,n\,x\,Sn - 3\,x\,Sn + n + 1]$$

```
> Snfree := select(f->not has(f,Sn),GBD);
```

$$Snfree := [-2\,x\,Dx\,n - x^2\,n + n + Dx^2\,n + 2\,n^2 - 2\,x^2\,Dx^2\,n - 2\,x\,Dx + Dx^2 + n^3$$
$$- 2\,Dx^2\,x^2 + 2\,Dx\,x^3 - 2\,x^2\,n^2 + 2\,Dx\,x^3\,n + Dx^2\,x^4\,n + Dx^2\,x^4 - x^2\,n^3]$$

```
> nops(Snfree);
```

$$1$$

```
> factor(Snfree[1]);
```

$$(n+1)\,(x-1)\,(x+1)\,(-Dx^2 + Dx^2\,x^2 + 2\,x\,Dx - n^2 - n)$$

```
> select(has,",Dx);
```

$$-Dx^2 + Dx^2\,x^2 + 2\,x\,Dx - n^2 - n$$

be careful doing the above command; it only works correctly if the expression is a product of >1 factors. If it had just ONE factor, the above expression would be a sum, not a product, so the selection would choose monomials with Dx in the sum.

```
> collect(",Dx,factor);
```

$$(x-1)(x+1)Dx^2+2\,x\,Dx-n\,(n+1)$$

```
> sort(",[n,x,Sn,Dx],plex);
```

$$-(n+1)\,n+2\,x\,Dx+(x-1)(x+1)\,Dx^2$$

And this is the negative of DE!

```
>
```

Page 48: Legendre polynomials, continued. Find a generating function for the Legendre polynomials P[n](x), using these annihilators:

```
> DE; RDE; RE;
```

$$(1-x^2)\,Dx^2-2\,x\,Dx+(n+1)\,n$$

$$(1-x^2)\,Dx\,Sn+(n+1)\,x\,Sn-n-1$$

$$(n+2)\,Sn^2-(2\,n+3)\,x\,Sn+n+1$$

The generating function is F(x,y) = sum_n P[n](x) * y^n.
Annihilators of P[n](x)*y^n are similar to ones for P[n](x)
each annihilator C(Sn,n,Dx,x) of P[n](x) gives an annihilator C(Sn/y,n,Dx,x) for P[n](x) * y^n.
Also, Dy * y^n = n*y^(n-1) so (y Dy - n) annihilates y^n without touching P[n](x).

```
> G := map(expand,
        [DE,
         numer(normal(subs(Sn=Sn/y,RE))),
         numer(normal(subs(Sn=Sn/y,RDE))),
         y*Dy-n]);
```

$$G := [Dx^2-Dx^2\,x^2-2\,x\,Dx+n^2+n,\ n\,Sn^2+2\,Sn^2-2\,x\,Sn\,y\,n-3\,x\,Sn\,y+n\,y^2+y^2,$$
$$Dx\,Sn-Dx\,Sn\,x^2+n\,x\,Sn+x\,Sn-n\,y-y,\ y\,Dy-n]$$

```
> A :=
  skew_algebra(shift=[Sn,n],diff=[Dx,x],diff=[Dy,y],polyno
  m=[n,x,y]);
```

$$A := Ore\_algebra$$

The sum is over n, so find n-free recurrence/diffeqs for the summand (just like it's k-free when the sum is over k):

```
> T := termorder(A,lexdeg([n],[x,Dx,y,Dy,Sn]));
```

$$T := term\_order$$

```
> GN := gbasis(G,T);
```

$$GN := [-x\,Sn\,y\,Dy-Dx\,Sn+Dx\,Sn\,x^2+y+y^2\,Dy,$$
$$-Dx^2+Dx^2\,x^2+2\,x\,Dx-y^2\,Dy^2-2\,y\,Dy,$$
$$x\,Sn\,y-y^2+2\,x\,Sn\,y^2\,Dy-y^3\,Dy-Sn^2\,y\,Dy,$$
$$-Sn\,y^2\,Dy^2-Sn\,y\,Dy+Dx\,x\,Sn\,y\,Dy-Dx\,y-Dx\,y^2\,Dy,$$

$$x\,Dx\,y^2\,Dy + y^3\,Dy^2 + 3\,y^2\,Dy + Dx\,x\,y - Dx\,Sn\,y\,Dy + y,$$
$$Dx\,x\,Sn\,y + Dx\,y^2 + Dx\,y^3\,Dy - Dx\,Sn^2\,y\,Dy + Sn\,y + 4\,Sn\,y^2\,Dy + 2\,Sn\,y^3\,Dy^2,$$
$$-y\,Dy + n\,]$$

Get all the n-free recurrence/diffeqs this produced:

```
> SN := select((p,v)-> not has(p,v), GN, n);
```

$$SN := [-x\,Sn\,y\,Dy - Dx\,Sn + Dx\,Sn\,x^2 + y + y^2\,Dy,$$
$$-Dx^2 + Dx^2\,x^2 + 2\,x\,Dx - y^2\,Dy^2 - 2\,y\,Dy,$$
$$x\,Sn\,y - y^2 + 2\,x\,Sn\,y^2\,Dy - y^3\,Dy - Sn^2\,y\,Dy,$$
$$-Sn\,y^2\,Dy^2 - Sn\,y\,Dy + Dx\,x\,Sn\,y\,Dy - Dx\,y - Dx\,y^2\,Dy,$$
$$x\,Dx\,y^2\,Dy + y^3\,Dy^2 + 3\,y^2\,Dy + Dx\,x\,y - Dx\,Sn\,y\,Dy + y,$$
$$Dx\,x\,Sn\,y + Dx\,y^2 + Dx\,y^3\,Dy - Dx\,Sn^2\,y\,Dy + Sn\,y + 4\,Sn\,y^2\,Dy + 2\,Sn\,y^3\,Dy^2\,]$$

Recurrence/diffeqs for the sum are obtained from one for the summand as before by setting Sn=1 instead of Sk=1:

```
> ON := subs(Sn=1,SN);
```

$$ON := [-x\,y\,Dy - Dx + Dx\,x^2 + y + y^2\,Dy,\ -Dx^2 + Dx^2\,x^2 + 2\,x\,Dx - y^2\,Dy^2 - 2\,y\,Dy,$$
$$x\,y - y^2 + 2\,x\,y^2\,Dy - y^3\,Dy - y\,Dy,\ -y^2\,Dy^2 - y\,Dy + x\,Dx\,y\,Dy - Dx\,y - Dx\,y^2\,Dy,$$
$$x\,Dx\,y^2\,Dy + y^3\,Dy^2 + 3\,y^2\,Dy + Dx\,x\,y - Dx\,y\,Dy + y,$$
$$Dx\,x\,y + Dx\,y^2 + Dx\,y^3\,Dy - Dx\,y\,Dy + y + 4\,y^2\,Dy + 2\,y^3\,Dy^2\,]$$

```
> Axy := skew_algebra(diff=[Dx,x],diff=[Dy,y]);
  Txy := termorder(Axy,tdeg(Dx,Dy));
```

$$Axy := Ore\_algebra$$

$$Txy := term\_order$$

Since we know the generating function for the Legendre polynomials, we may test it on them. Get simplified equations:

```
> GF := gbasis(ON,Txy,Axy);
```

$$GF := [-y^2\,Dy - y + 2\,x\,y\,Dy + x - Dy,\ 2\,Dx\,x\,y - Dx - Dx\,y^2 + y\,]$$

```
> fn := (1-2*x*y+y^2)^(-1/2);
```

$$fn := \frac{1}{\sqrt{y^2 + 1 - 2\,x\,y}}$$

```
> map(oper -> applyopr(oper,fn,Axy),GF);
```

$$\left[\frac{x-y}{\sqrt{y^2+1-2\,x\,y}} - \frac{1}{2}\,\frac{(2\,x\,y - y^2 - 1)\,(2\,y - 2\,x)}{(y^2+1-2\,x\,y)^{3/2}},\right.$$

$$\frac{y}{\sqrt{y^2+1-2\,x\,y}} + \frac{(2\,x\,y - y^2 - 1)\,y}{(y^2+1-2\,x\,y)^{3/2}}\Bigg]$$

```
> normal(");
```

$$[0, 0]$$

Or if we didn't know the g.f. we could solve for it:

```
> eqx := applyopr(GF[2],f[y](x),Axy);
  eqy := applyopr(GF[1],f[x](y),Axy);
```

$$eqx := y\,f_y(x) + (2\,x\,y - y^2 - 1)\left(\frac{\partial}{\partial x} f_y(x)\right)$$

$$eqy := (x - y)\,f_x(y) + (2\,x\,y - y^2 - 1)\left(\frac{\partial}{\partial y} f_x(y)\right)$$

```
> dsolve(eqx,f[y](x));
```

$$f_y(x) = \frac{\_C1}{\sqrt{2\,x\,y - y^2 - 1}}$$

A priori, the constant _C1 is constant w.r.t. x only:

```
> subs(_C1=c(y),");
```

$$f_y(x) = \frac{\mathrm{c}(y)}{\sqrt{2\,x\,y - y^2 - 1}}$$

```
> subs(_C1=d(x),dsolve(eqy,f[x](y)));
```

$$f_x(y) = \frac{\mathrm{d}(x)}{\sqrt{2\,x\,y - y^2 - 1}}$$

c(y), d(x) are reconciled by having them both be independent of x,y, i.e., just a single numeric constant. The recurrence & diffeqs we began with do not contain any information on initial conditions, so we cannot say what this constant is without additional information.